// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.3) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    /**
     * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    /**
     * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
     * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
     */
    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    /**
     * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 oldAllowance = token.allowance(address(this), spender);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
    }

    /**
     * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
        }
    }

    /**
     * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
     * to be set to zero before setting it to a non-zero value, such as USDT.
     */
    function forceApprove(IERC20 token, address spender, uint256 value) internal {
        bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);

        if (!_callOptionalReturnBool(token, approvalCall)) {
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
            _callOptionalReturn(token, approvalCall);
        }
    }

    /**
     * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
     * Revert on invalid signature.
     */
    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     *
     * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
     */
    function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
        // and not revert is the subcall reverts.

        (bool success, bytes memory returndata) = address(token).call(data);
        return
            success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/structs/EnumerableSet.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/structs/EnumerableSet.sol)
// This file was procedurally generated from scripts/generate/templates/EnumerableSet.js.

pragma solidity ^0.8.0;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```solidity
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 *
 * [WARNING]
 * ====
 * Trying to delete such a structure from storage will likely result in data corruption, rendering the structure
 * unusable.
 * See https://github.com/ethereum/solidity/pull/11843[ethereum/solidity#11843] for more info.
 *
 * In order to clean an EnumerableSet, you can either remove all elements one by one or create a fresh instance using an
 * array of EnumerableSet.
 * ====
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;
        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping(bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) {
            // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            if (lastIndex != toDeleteIndex) {
                bytes32 lastValue = set._values[lastIndex];

                // Move the last value to the index where the value to delete is
                set._values[toDeleteIndex] = lastValue;
                // Update the index for the moved value
                set._indexes[lastValue] = valueIndex; // Replace lastValue's index to valueIndex
            }

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        return set._values[index];
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function _values(Set storage set) private view returns (bytes32[] memory) {
        return set._values;
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(Bytes32Set storage set) internal view returns (bytes32[] memory) {
        bytes32[] memory store = _values(set._inner);
        bytes32[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint160(uint256(_at(set._inner, index))));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(AddressSet storage set) internal view returns (address[] memory) {
        bytes32[] memory store = _values(set._inner);
        address[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(UintSet storage set) internal view returns (uint256[] memory) {
        bytes32[] memory store = _values(set._inner);
        uint256[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }
}

// File: contracts/v2/BaseVault.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Address.sol";
import "@openzeppelin/contracts/utils/structs/EnumerableSet.sol";

import "./interfaces/INftVault.sol";

/// @notice Base configuration contract for the NFT Vault
abstract contract BaseVault is INftVault, Ownable {
    using Address for address payable;
    using EnumerableSet for EnumerableSet.AddressSet;
    using EnumerableSet for EnumerableSet.UintSet;

    /// @notice Basis points denominator in fee values
    uint256 private constant BASIS_POINTS_DENOMINATOR = 10_000;

    /// @notice max lock time buffer
    uint256 internal constant MAX_LOCK_TIME_BUFFER = 1825 days; // 5 years

    /// @notice max deposit fee 10%
    uint16 private constant MAX_DEPOSIT_FEE = 1000;
    /// @notice max withdraw fee 10%
    uint16 private constant MAX_WITHDRAW_FEE = 1000;

    /// @notice Array limit in the batch tx functions. Default 20 items
    uint256 internal _batchTxLimit = 20;

    /// @notice deposit fee percentage. denominator 10000
    uint256 internal _depositFee;
    /// @notice Withdraw fee percentage. denominator 10000
    uint256 internal _withdrawFee;

    /// @notice Flag to see if we use only depositable nfts (true) or use all nfts (false)
    bool internal _isUseNftWhitelist;

    /// @notice deposit fee receive wallet address
    address payable internal _treasury;

    address internal _depositValidator;
    address internal _withdrawValidator;

    /// @notice ERC20 tokens which are exempted from the deposit fee when deposited
    mapping(address => bool) internal _depositFeeExemptedTokenList;

    /// @notice ERC20 tokens which are exempted from the withdraw fee when withdrawn
    mapping(address => bool) internal _withdrawFeeExemptedTokenList;

    /// @notice A wrapper is allowed to withdraw user's token deposited
    /// @dev key is a hash of user address and wrapper address
    mapping(bytes32 => bool) internal _withdrawForPaymentApproved;
    /// @notice Payment wrapper contract list which can make payment for Slash
    EnumerableSet.AddressSet internal _wrapperList;

    /// @notice Allowed nfts to be used in the NFT Vault
    EnumerableSet.AddressSet internal _nftWhitelist;

    /// @notice NFT address is on the blacklist
    EnumerableSet.AddressSet internal _nftBlacklist;

    /// @notice Tokenid is on the blacklist
    mapping(address => EnumerableSet.UintSet) internal _tokenIdBlacklist;

    /// @notice EIP20 address is on the blacklist
    EnumerableSet.AddressSet internal _tokenBlacklist;

    /// @notice Update limit of the array in the batch tx functions
    /// @dev Only owner is allowed to call this function
    function updateBatchTxLimit(uint256 limit_) external onlyOwner {
        if (_batchTxLimit == limit_) revert AlreadyConfigured();
        _batchTxLimit = limit_;
        emit BatchTxLimitUpdated(limit_);
    }

    function batchTxLimit() external view returns (uint256) {
        return _batchTxLimit;
    }

    /// @notice Update treasury wallet to receive fee
    /// @param treasury_ fee receive address
    function updateTreasury(address payable treasury_) external onlyOwner {
        if (treasury_ == address(0)) revert InvalidZeroAddress();
        if (_treasury == treasury_) revert AlreadyConfigured();

        treasury_.sendValue(0); // We try to send 0 ETH to the new treasury address to check if it can receive ETH

        _treasury = treasury_;

        emit TreasuryUpdated(treasury_);
    }

    function treasury() external view returns (address payable) {
        return _treasury;
    }

    /// @notice Exempt token from the deposit fee
    function exemptTokenFromDepositFee(
        address token_,
        bool flag_
    ) external onlyOwner {
        if (_depositFeeExemptedTokenList[token_] == flag_)
            revert AlreadyConfigured();
        _depositFeeExemptedTokenList[token_] = flag_;

        emit TokenExemptedFromDepositFee(token_, flag_);
    }

    /// @notice Check if the token is exempted from deposit fee in the vault
    function isTokenExemptedFromDepositFee(
        address token_
    ) external view returns (bool) {
        return _depositFeeExemptedTokenList[token_];
    }

    /// @notice Get fee amount when deposit is made with the `token` and `amount`
    /// @dev For the exempted token, fee amount is 0
    /// @param token_ ERC20 token which is being deposited
    function depositFee(
        address token_,
        uint256 amount_
    ) public view override returns (uint256) {
        if (_depositFeeExemptedTokenList[token_]) return 0;
        return (amount_ * _depositFee) / BASIS_POINTS_DENOMINATOR;
    }

    function depositFee() external view returns (uint256) {
        return _depositFee;
    }

    /// @notice Update deposit fee percentage
    /// @dev Must be less than MAX_DEPOSIT_FEE 10%
    /// @param fee_ deposit fee percentage (denominator 10000)
    function updateDepositFee(uint256 fee_) external onlyOwner {
        if (fee_ > MAX_DEPOSIT_FEE) revert Overflow256(MAX_DEPOSIT_FEE, fee_);
        if (_depositFee == fee_) revert AlreadyConfigured();
        _depositFee = fee_;
        emit DepositFeeUpdated(fee_);
    }

    /// @notice Exempt token from the withdraw fee
    function exemptTokenFromWithdrawFee(
        address token_,
        bool flag_
    ) external onlyOwner {
        if (_withdrawFeeExemptedTokenList[token_] == flag_)
            revert AlreadyConfigured();
        _withdrawFeeExemptedTokenList[token_] = flag_;

        emit TokenExemptedFromWithdrawFee(token_, flag_);
    }

    /// @notice Check if the token is exempted from deposit fee in the vault
    function isTokenExemptedFromWithdrawFee(
        address token_
    ) external view returns (bool) {
        return _withdrawFeeExemptedTokenList[token_];
    }

    /// @notice Update withdraw fee percentage
    /// @dev Must be less than MAX_WITHDRAW_FEE 10%
    /// @param fee_ withdraw fee percentage (denominator 10000)
    function updateWithdrawFee(uint256 fee_) external onlyOwner {
        if (fee_ > MAX_WITHDRAW_FEE) revert Overflow256(MAX_WITHDRAW_FEE, fee_);
        if (_withdrawFee == fee_) revert AlreadyConfigured();
        _withdrawFee = fee_;

        emit WithdrawFeeUpdated(fee_);
    }

    /// @notice Get fee amount when withdraw is made with the `token` and `amount`
    /// @dev For the exempted token, fee amount is 0
    /// @param token_ ERC20 token which is being withdrawn
    function withdrawFee(
        address token_,
        uint256 amount_
    ) public view override returns (uint256) {
        if (_withdrawFeeExemptedTokenList[token_]) return 0;
        return (amount_ * _withdrawFee) / BASIS_POINTS_DENOMINATOR;
    }

    function withdrawFee() external view returns (uint256) {
        return _withdrawFee;
    }

    /// @notice Update deposit validator
    function updateDepositValidator(address validator_) external onlyOwner {
        if (_depositValidator == validator_) revert AlreadyConfigured();
        if (validator_ == address(0)) revert InvalidZeroAddress();
        _depositValidator = validator_;
    }

    function depositValidator() external view returns (address) {
        return _depositValidator;
    }

    /// @notice Update withdraw validator
    function updateWithdrawValidator(address validator_) external onlyOwner {
        if (_withdrawValidator == validator_) revert AlreadyConfigured();
        if (validator_ == address(0)) revert InvalidZeroAddress();

        _withdrawValidator = validator_;
    }

    function withdrawValidator() external view returns (address) {
        return _withdrawValidator;
    }

    /// @notice Check if the NFT deposit whitelist is being used
    /// @return - true when used, false when not being used
    function isUseNftWhitelist() external view returns (bool) {
        return _isUseNftWhitelist;
    }

    /// @notice Set the flag for using deposit whitelist or not
    function updateIsUseNftWhitelist(bool flag_) external onlyOwner {
        if (_isUseNftWhitelist == flag_) revert AlreadyConfigured();
        _isUseNftWhitelist = flag_;

        emit NftWhitelistUsed(flag_);
    }

    /// @notice Get whitelisted nft count
    function whitelistedNftCount() external view returns (uint256) {
        return _nftWhitelist.length();
    }

    /// @notice Get blacklisted tokenId
    function blacklistTokenIdCount(
        address nftAddress_
    ) external view returns (uint256) {
        return _tokenIdBlacklist[nftAddress_].length();
    }

    /// @notice Get blacklisted nft
    function blacklistNftCount() external view returns (uint256) {
        return _nftBlacklist.length();
    }

    /// @notice Get blacklisted token
    function blacklistTokenCount() external view returns (uint256) {
        return _tokenBlacklist.length();
    }

    /// @notice Allow / disallow the given nft address for the deposit
    /// @dev We always update the whitelist regardless whitelist is used or not
    function updateNftWhitelist(
        address[] calldata nftAddresses_,
        bool flag_
    ) external onlyOwner {
        uint256 i;
        uint256 length = nftAddresses_.length;
        for (; i < length; ) {
            if (_nftWhitelist.contains(nftAddresses_[i]) != flag_) {
                if (flag_) _nftWhitelist.add(nftAddresses_[i]);
                else _nftWhitelist.remove(nftAddresses_[i]);
            }

            unchecked {
                ++i;
            }
        }

        emit NftWhitelistUpdated(nftAddresses_, flag_);
    }

    /// @notice Allow / disallow the given nft address for the deposit
    /// @dev We always update the whitelist regardless whitelist is used or not
    function updateNftBlacklist(
        address[] calldata nftAddresses_,
        bool flag_
    ) external onlyOwner {
        uint256 i;
        uint256 length = nftAddresses_.length;
        for (; i < length; ) {
            if (_nftBlacklist.contains(nftAddresses_[i]) != flag_) {
                if (flag_) _nftBlacklist.add(nftAddresses_[i]);
                else _nftBlacklist.remove(nftAddresses_[i]);
            }

            unchecked {
                ++i;
            }
        }

        emit NftBlacklistUpdated(nftAddresses_, flag_);
    }

    /// @notice Allow / disallow the given token address for the deposit
    /// @dev We always update the blacklist regardless blacklist is used or not
    function updateTokenBlacklist(
        address[] calldata tokens_,
        bool flag_
    ) external onlyOwner {
        uint256 i;
        uint256 length = tokens_.length;
        for (; i < length; ) {
            if (_tokenBlacklist.contains(tokens_[i]) != flag_) {
                if (flag_) _tokenBlacklist.add(tokens_[i]);
                else _tokenBlacklist.remove(tokens_[i]);
            }

            unchecked {
                ++i;
            }
        }

        emit TokenBlacklistUpdated(tokens_, flag_);
    }

    /// @notice Add / remove the exact tokenId in nft collection for the blacklist deposit
    function updateTokenIdBlacklist(
        address nftAddress_,
        uint256 tokenId_,
        bool flag_
    ) external onlyOwner {
        if (_tokenIdBlacklist[nftAddress_].contains(tokenId_) == flag_)
            revert AlreadyConfigured();

        if (flag_) {
            _tokenIdBlacklist[nftAddress_].add(tokenId_);
            if (!_nftBlacklist.contains(nftAddress_))
                _nftBlacklist.add(nftAddress_);
        } else {
            _tokenIdBlacklist[nftAddress_].remove(tokenId_);
            if (_tokenIdBlacklist[nftAddress_].length() == 0)
                _nftBlacklist.remove(nftAddress_);
        }

        emit TokenIdBlacklistUpdated(nftAddress_, tokenId_, flag_);
    }

    /// @notice Check tokenId is blacklisted or not
    function tokenIdInBlacklist(
        address nftAddress_,
        uint256 tokenId_
    ) public view returns (bool) {
        return _tokenIdBlacklist[nftAddress_].contains(tokenId_);
    }

    /// @notice Check token is blacklisted or not
    function tokenInBlacklist(address token) public view returns (bool) {
        return _tokenBlacklist.contains(token);
    }

    /// @notice Return deposit allowed nft list
    /// @dev This function is paginated for considering gas wastes in bulk cases
    function getNftWhitelistPaging(
        uint256 offset_,
        uint256 limit_
    )
        external
        view
        returns (
            address[] memory nftAddresses,
            uint256[] memory nftTokenIds,
            uint256 nextOffset,
            uint256 total
        )
    {
        EnumerableSet.AddressSet storage nftWhitelist = _nftWhitelist;
        total = nftWhitelist.length();
        nftTokenIds = new uint256[](0); // nftTokenIds will not be used at the moment
        if (offset_ >= total) nextOffset = offset_;
        else {
            if (offset_ + limit_ > total) limit_ = total - offset_;
            nextOffset = offset_ + limit_;
            nftAddresses = new address[](limit_);
            uint256 i;
            for (; i < limit_; ) {
                nftAddresses[i] = nftWhitelist.at(i + offset_);
                unchecked {
                    ++i;
                }
            }
        }
    }

    /// @notice Return deposit blacklist nft list
    /// @dev This function is paginated for considering gas wastes in bulk cases
    function getNftBlacklistPaging(
        uint256 offset_,
        uint256 limit_
    )
        external
        view
        returns (
            address[] memory nftAddresses,
            uint256 nextOffset,
            uint256 total
        )
    {
        EnumerableSet.AddressSet storage nftBlacklist = _nftBlacklist;
        total = nftBlacklist.length();
        if (offset_ >= total) nextOffset = offset_;
        else {
            if (offset_ + limit_ > total) limit_ = total - offset_;
            nextOffset = offset_ + limit_;
            nftAddresses = new address[](limit_);
            uint256 i;
            for (; i < limit_; ) {
                nftAddresses[i] = nftBlacklist.at(i + offset_);
                unchecked {
                    ++i;
                }
            }
        }
    }

    /// @notice Return deposit blacklist token list
    /// @dev This function is paginated for considering gas wastes in bulk cases
    function getTokenBlacklistPaging(
        uint256 offset_,
        uint256 limit_
    )
        external
        view
        returns (address[] memory tokens, uint256 nextOffset, uint256 total)
    {
        EnumerableSet.AddressSet storage tokenBlacklist = _tokenBlacklist;
        total = tokenBlacklist.length();
        if (offset_ >= total) nextOffset = offset_;
        else {
            if (offset_ + limit_ > total) limit_ = total - offset_;
            nextOffset = offset_ + limit_;
            tokens = new address[](limit_);
            uint256 i;
            for (; i < limit_; ) {
                tokens[i] = tokenBlacklist.at(i + offset_);
                unchecked {
                    ++i;
                }
            }
        }
    }

    /// @notice Return deposit blacklist nft tokenid
    /// @dev This function is paginated for considering gas wastes in bulk cases
    function getTokenIdBlacklistPaging(
        address nftAddress_,
        uint256 offset_,
        uint256 limit_
    )
        external
        view
        returns (
            uint256[] memory nftTokenIds,
            uint256 nextOffset,
            uint256 total
        )
    {
        EnumerableSet.UintSet storage tokenIdBlacklist = _tokenIdBlacklist[
            nftAddress_
        ];
        total = tokenIdBlacklist.length();
        nftTokenIds = new uint256[](0); // nftTokenIds will not be used at the moment
        if (offset_ >= total) nextOffset = offset_;
        else {
            if (offset_ + limit_ > total) limit_ = total - offset_;
            nextOffset = offset_ + limit_;
            nftTokenIds = new uint256[](limit_);
            uint256 i;
            for (; i < limit_; ) {
                nftTokenIds[i] = tokenIdBlacklist.at(i + offset_);
                unchecked {
                    ++i;
                }
            }
        }
    }

    /// @notice Get payment wrapper count
    function wrapperCount() external view returns (uint256) {
        return _wrapperList.length();
    }

    /// @notice Return wrapper count
    /// @dev This function is paginated for considering gas wastes in bulk cases
    function getWrapperListPaging(
        uint256 offset_,
        uint256 limit_
    )
        external
        view
        returns (address[] memory wrappers, uint256 nextOffset, uint256 total)
    {
        EnumerableSet.AddressSet storage wrapperList = _wrapperList;
        total = wrapperList.length();
        if (offset_ >= total) nextOffset = offset_;
        else {
            if (offset_ + limit_ > total) limit_ = total - offset_;
            nextOffset = offset_ + limit_;
            wrappers = new address[](limit_);
            uint256 i;
            for (; i < limit_; ) {
                wrappers[i] = wrapperList.at(i + offset_);
                unchecked {
                    ++i;
                }
            }
        }
    }

    /// @notice Add / remove item in the wrapper list
    function updateWrapperList(
        address wrapper_,
        bool flag_
    ) external onlyOwner {
        if (_wrapperList.contains(wrapper_) == flag_)
            revert AlreadyConfigured();

        if (flag_) _wrapperList.add(wrapper_);
        else _wrapperList.remove(wrapper_);

        emit WrapperUpdated(wrapper_, flag_);
    }

    /// @notice Check if the given contract is allowed to call deposit & payment function
    function isWrapper(address contract_) external view returns (bool) {
        return _wrapperList.contains(contract_);
    }

    /// @notice Approve wrapper to withdraw funds from the vault contract
    function approveWrapper(address wrapper_, bool flag_) external {
        if (!_wrapperList.contains(wrapper_) && flag_) revert InvalidWrapper();
        bytes32 key = keccak256(abi.encodePacked(_msgSender(), wrapper_));
        if (_withdrawForPaymentApproved[key] == flag_)
            revert AlreadyConfigured();
        _withdrawForPaymentApproved[key] = flag_;

        emit WrapperApprovedForPayment(_msgSender(), wrapper_, flag_);
    }

    /// @notice Check if the wrapper is approved to withdraw user's tokens
    function wrapperApproved(
        address account_,
        address wrapper_
    ) external view returns (bool) {
        bytes32 key = keccak256(abi.encodePacked(account_, wrapper_));
        return _withdrawForPaymentApproved[key];
    }
}

// File: contracts/v2/interfaces/IDepositValidator.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

interface IDepositValidator {
    /// @notice Check if the given nft is valid one for being used in nft vault
    /// @dev nftAddress is only used at the moment, other args may be used later
    /// @dev Only ERC721 standard is being supported at the moment
    function isValid(
        address /* remitter_ */, // may be used later
        address nftAddress_,
        uint256 /* nftTokenId_ */ // may be used later
    ) external view returns (bool);
}

// File: contracts/v2/interfaces/INftVault.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

interface INftVault {
    enum BalanceFetchType {
        UNLOCKED_ONLY,
        LOCKED_ONLY,
        ALL
    }

    /// @notice Get fee amount when withdraw is made with the `token` and `amount`
    /// @dev For the exempted token, fee amount is 0
    /// @param token_ ERC20 token which is being withdrawn
    function withdrawFee(
        address token_,
        uint256 amount_
    ) external view returns (uint256);

    /// @notice Get fee amount when deposit is made with the `token` and `amount`
    /// @dev For the exempted token, fee amount is 0
    /// @param token_ ERC20 token which is being deposited
    function depositFee(
        address token_,
        uint256 amount_
    ) external view returns (uint256);

    /// @notice Batch deposit for several nft tokens + token ids
    /// @param erc20Token_ ERC20 token contract address to deposit
    /// @param nftAddresses_ key NFT contract addresses
    /// @param nftTokenIds_ key NFT contract token IDs
    /// @param amounts_ deposit amounts
    /// @param unlockAt_ the time until the deposited tokens are locked
    function batchDeposit(
        address erc20Token_,
        address[] calldata nftAddresses_,
        uint256[] calldata nftTokenIds_,
        uint256[] calldata amounts_,
        uint64 unlockAt_
    ) external payable;

    /// @notice Withdraw for making payment on the Slash Protocol
    /// @param account_ Account who is going to make payment via Slash
    /// @param nftAddress_ key NFT contract address
    /// @param nftTokenId_ key NFT token ID
    /// @param erc20Token_ withdraw token contract address
    /// @param amount_ withdraw amount
    function withdrawForPayment(
        address account_,
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_
    ) external;

    event BatchTxLimitUpdated(uint256 limit);
    event Deposited(
        address erc20Token,
        address nftAddress,
        uint256 nftTokenId,
        uint256 amount,
        uint64 unlockAt
    );
    event DepositFeeUpdated(uint256 fee);
    event NftWhitelistUpdated(address nftAddress, bool flag);
    event TokenIdBlacklistUpdated(
        address nftAddress,
        uint256 tokenId,
        bool flag
    );
    event UpdateTrustedForwarder(address forwarder);
    event NftWhitelistUpdated(address[] nftAddress, bool flag);
    event NftBlacklistUpdated(address[] nftAddress, bool flag);
    event TokenBlacklistUpdated(address[] tokens, bool flag);
    event NftWhitelistUsed(bool flag);
    event RecoverWrongToken(address token, uint256 amount);
    event TokenExemptedFromDepositFee(address token, bool flag);
    event TokenExemptedFromWithdrawFee(address token, bool flag);
    event TreasuryUpdated(address payable treasury);
    event Locked(
        address holder,
        address erc20Token,
        address nftAddress,
        uint256 nftTokenId,
        uint256 amount,
        uint64 unlockAt
    );
    event Relocked(
        address holder,
        address erc20Token,
        address nftAddress,
        uint256 nftTokenId,
        uint64 lockId,
        uint64 unlockAt
    );
    event Unlocked(
        address holder,
        address erc20Token,
        address nftAddress,
        uint256 nftTokenId,
        uint64 lockId
    );
    event WithdrawFeeUpdated(uint256 fee);
    event Withdrawn(
        address recipient,
        address erc20Token,
        address nftAddress,
        uint256 nftTokenId,
        uint256 amountWithdrawn,
        uint256 feePaid
    );
    event WrapperApprovedForPayment(
        address account,
        address wrapper,
        bool flag
    );
    event WrapperUpdated(address wrapper, bool flag);

    error AlreadyConfigured();
    error AlreadyLocked();
    error InsufficientTransfer(uint256 expected, uint256 actual);
    error InvalidNft();
    error InvalidWrapper();
    error InvalidZeroAddress();
    error InvalidZeroAmount();
    error Overflow256(uint256 limit, uint256 actual);
    error TooMuchWithdrawals(uint256 available, uint256 actual);
    error UnapprovedWrapper();
    error Unpermitted();
    error InvalidUnlockTime();
    error InvalidArraySize();
    error InvalidToken(address token);
}

// File: contracts/v2/interfaces/IWithdrawValidator.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

interface IWithdrawValidator {
    /// @notice Check if the owner is able to withdraw tokens from nft vault
    /// @dev It just checks if the owner holds given `tokenId` of the `nftAddress`
    /// @dev Only ERC721 standard is being supported at the moment
    function isValid(
        address owner_,
        address nftAddress_,
        uint256 tokenId_
    ) external view returns (bool);
}

// File: contracts/v2/libs/IterableAddressMap.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

struct ItAddressMap {
    // address => index
    mapping(address => uint256) indexs;
    // array of address
    address[] keys;
    // never use it, just for keep compile success.
    uint256 size;
}

library IterableAddressMap {
    /// @notice Insert new address
    function insert(ItAddressMap storage self_, address key_) internal {
        uint256 keyIndex = self_.indexs[key_];
        if (keyIndex > 0) return;
        else {
            self_.keys.push(key_);
            self_.indexs[key_] = self_.keys.length;
        }
    }

    /// @notice Remove address
    function remove(ItAddressMap storage self_, address key_) internal {
        uint256 index = self_.indexs[key_];
        if (index == 0) return;
        address lastKey = self_.keys[self_.keys.length - 1];
        if (key_ != lastKey) {
            self_.keys[index - 1] = lastKey;
            self_.indexs[lastKey] = index;
        }
        delete self_.indexs[key_];
        self_.keys.pop();
    }

    /// @notice Check if the address is contained
    function contains(
        ItAddressMap storage self_,
        address key_
    ) internal view returns (bool) {
        return self_.indexs[key_] > 0;
    }

    /// @notice View the count of addresses added to this struct
    function itemCount(
        ItAddressMap storage self_
    ) internal view returns (uint256) {
        return self_.keys.length;
    }

    /// @notice View addresses added to this struct
    /// @dev This function is paginated for supporting the bulk addresses
    function fetchItems(
        ItAddressMap storage self_,
        uint256 offset_,
        uint256 count_
    ) internal view returns (address[] memory paginatedKeys) {
        address[] memory keys = self_.keys;
        if (offset_ < keys.length) {
            if (offset_ + count_ > keys.length) count_ = keys.length - offset_;
            paginatedKeys = new address[](count_);
            uint256 i;
            for (; i < count_; i++) paginatedKeys[i] = keys[i + offset_];
        }
    }
}

// File: contracts/v2/libs/IterableLock.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

struct ItLock {
    // unlock time => amount
    mapping(uint64 => uint256) data;
    // unlock time => index
    mapping(uint64 => uint256) indexs;
    // array of unlock time
    uint64[] keys;
    // unlocked amount
    uint256 unlockedAmount;
    // never use it, just for keep compile success.
    uint256 size;
}

library IterableLock {
    error InsufficientAmount();
    error InvalidKey();
    error NotExpired();
    error NothingChanged();
    error OnlyExtendable();
    error OnlyLockable();
    error TooMuchAmount();

    ///@notice Deposit new lock to this struct
    function deposit(
        ItLock storage self_,
        uint64 key_,
        uint256 value_
    ) internal {
        if (value_ == 0) revert NothingChanged();
        if (key_ <= block.timestamp) {
            self_.unlockedAmount += value_;
            return;
        }

        uint256 keyIndex = self_.indexs[key_];
        self_.data[key_] += value_; // value is added to the existing unlock time
        if (keyIndex > 0) return;
        // When the key not exists, add it
        self_.keys.push(key_);
        self_.indexs[key_] = self_.keys.length;
    }

    /// @notice Lock with `key` from the `unlockedAmount`
    function lock(ItLock storage self_, uint64 key_, uint256 value_) internal {
        if (key_ <= block.timestamp) revert OnlyLockable();
        uint256 availableValue = self_.unlockedAmount;
        if (value_ > availableValue) revert TooMuchAmount();

        self_.unlockedAmount -= value_;

        uint256 keyIndex = self_.indexs[key_];
        self_.data[key_] += value_; // value is added to the existing unlock time
        if (keyIndex > 0) return;
        // When the key not exists, add it
        self_.keys.push(key_);
        self_.indexs[key_] = self_.keys.length;
    }

    /// @notice Relock from `key` to `newKey`
    function relock(
        ItLock storage self_,
        uint64 key_,
        uint64 newKey_
    ) internal {
        if (key_ >= newKey_) revert OnlyExtendable();
        uint256 keyIndex = self_.indexs[key_];
        if (keyIndex == 0) revert InvalidKey();

        uint256 lockedAmount = self_.data[key_];

        // To remove the old key, key is swapped with last key in the key array, and then pop the `last key` which is `key` now
        // Update index for the swapped `last key`, and then remove index and data for the `removing key`
        uint64 lastKey = self_.keys[self_.keys.length - 1];
        if (key_ != lastKey) {
            self_.keys[keyIndex - 1] = lastKey;
            self_.indexs[lastKey] = keyIndex;
        }

        delete self_.data[key_];
        delete self_.indexs[key_];
        self_.keys.pop();

        // If new unlock time is before the current time, just add it to the unlocked amount
        if (newKey_ <= block.timestamp) {
            self_.unlockedAmount += lockedAmount;
            return;
        }
        // If new key does not exist in the key array, add it
        uint256 newKeyIndex = self_.indexs[newKey_];
        if (newKeyIndex == 0) {
            self_.keys.push(newKey_);
            self_.indexs[newKey_] = self_.keys.length;
        }
        self_.data[newKey_] += lockedAmount;
    }

    /// @notice Unlock the expired lock
    function unlock(ItLock storage self_, uint64 key_) internal {
        if (key_ > block.timestamp) revert NotExpired();
        uint256 keyIndex = self_.indexs[key_];
        if (keyIndex == 0) revert InvalidKey();
        self_.unlockedAmount += self_.data[key_];

        // To remove this key, key is swapped with last key in the key array, and then pop the `last key` which is `key` now
        // Update index for the swapped `last key`, and then remove index and data for the `removing key`
        uint64 lastKey = self_.keys[self_.keys.length - 1];
        if (key_ != lastKey) {
            self_.keys[keyIndex - 1] = lastKey;
            self_.indexs[lastKey] = keyIndex;
        }

        delete self_.data[key_];
        delete self_.indexs[key_];
        self_.keys.pop();
    }

    /// @notice Withdraw unlocked token
    function withdraw(ItLock storage self_, uint256 amount_) internal {
        if (self_.unlockedAmount < amount_) revert InsufficientAmount();
        self_.unlockedAmount -= amount_;
    }

    /// @notice Check if there is no lock for this token
    function empty(ItLock storage self_) internal view returns (bool) {
        return self_.keys.length == 0 && self_.unlockedAmount == 0;
    }

    /// @notice Get available amount (unlocked)
    function availableAmount(
        ItLock storage self_
    ) internal view returns (uint256) {
        return self_.unlockedAmount;
    }

    /// @notice Fetch locks
    /// @dev This function is pagniated for supporting the bulk locks
    function fetchItems(
        ItLock storage self_,
        uint256 offset_,
        uint256 count_
    )
        internal
        view
        returns (uint64[] memory unlockDates, uint256[] memory amounts)
    {
        uint64[] memory keys = self_.keys;
        if (offset_ < keys.length) {
            if (offset_ + count_ > keys.length) count_ = keys.length - offset_;
            unlockDates = new uint64[](count_);
            amounts = new uint256[](count_);
            uint256 i;
            for (; i < count_; i++) {
                uint64 key = keys[i + offset_];
                unlockDates[i] = key;
                amounts[i] = self_.data[key];
            }
        }
    }

    /// @notice Fetch locked amount from the key (unlock date)
    /// @dev Actually, `key_` is same as `unlockDate`, but we return it again to keep the style
    function fetchItem(
        ItLock storage self_,
        uint64 key_
    ) internal view returns (uint64 unlockDate, uint256 amount) {
        unlockDate = key_;
        uint256 keyIndex = self_.indexs[key_];
        if (keyIndex > 0) amount = self_.data[key_];
    }

    /// @notice Check if the deposited tokens exist for the key
    function exists(
        ItLock storage self_,
        uint64 key_
    ) internal view returns (bool) {
        uint256 keyIndex = self_.indexs[key_];
        if (keyIndex > 0) return self_.data[key_] > 0;
        return false;
    }
}

// File: contracts/v2/libs/UniversalERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.4;

import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/utils/Address.sol";

// File: contracts/libs/UniversalERC20.sol
/**
 * @notice Library for wrapping ERC20 token and ETH
 * @dev It uses msg.sender directly so only use in normal contract, not in GSN-like contract
 */
library UniversalERC20 {
    using SafeERC20 for IERC20;
    using Address for address payable;

    IERC20 internal constant ETH_ADDRESS =
        IERC20(0xEeeeeEeeeEeEeeEeEeEeeEEEeeeeEeeeeeeeEEeE);

    function universalTransfer(
        IERC20 token,
        address to,
        uint256 amount
    ) internal returns (uint256) {
        if (amount == 0) {
            return 0;
        }

        if (isETH(token)) {
            payable(address(uint160(to))).sendValue(amount);
            return amount;
        }
        uint256 balanceBefore = token.balanceOf(to);
        token.safeTransfer(to, amount);
        return token.balanceOf(to) - balanceBefore;
    }

    function universalTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 amount
    ) internal returns (uint256) {
        if (amount == 0) {
            return 0;
        }

        if (isETH(token)) {
            require(msg.value >= amount, "Insufficient msg.value");
            if (to != address(this))
                payable(address(uint160(to))).sendValue(amount);

            // refund redundant amount
            if (msg.value > amount)
                payable(address(uint160(from))).sendValue(msg.value - amount);

            return amount;
        }
        uint256 balanceBefore = token.balanceOf(to);
        token.safeTransferFrom(from, to, amount);
        return token.balanceOf(to) - balanceBefore;
    }

    function universalTransferFromSenderToThis(
        IERC20 token,
        uint256 amount
    ) internal returns (uint256) {
        if (amount == 0) {
            return 0;
        }

        if (isETH(token)) {
            require(msg.value >= amount, "Insufficient msg.value");
            // Return remainder if exist
            if (msg.value > amount)
                payable(msg.sender).sendValue(msg.value - amount);
            return amount;
        }
        uint256 balanceBefore = token.balanceOf(address(this));
        token.safeTransferFrom(msg.sender, address(this), amount);
        return token.balanceOf(address(this)) - balanceBefore;
    }

    function universalApprove(
        IERC20 token,
        address to,
        uint256 amount
    ) internal {
        if (!isETH(token)) {
            if (amount > 0 && token.allowance(address(this), to) > 0) {
                token.safeApprove(to, 0);
            }
            token.safeApprove(to, amount);
        }
    }

    function universalBalanceOf(
        IERC20 token,
        address who
    ) internal view returns (uint256) {
        if (isETH(token)) {
            return who.balance;
        } else {
            return token.balanceOf(who);
        }
    }

    function universalDecimals(IERC20 token) internal view returns (uint256) {
        if (isETH(token)) {
            return 18;
        }

        (bool success, bytes memory data) = address(token).staticcall{
            gas: 10000
        }(abi.encodeWithSignature("decimals()"));
        if (!success || data.length == 0) {
            (success, data) = address(token).staticcall{gas: 10000}(
                abi.encodeWithSignature("DECIMALS()")
            );
        }

        return (success && data.length > 0) ? abi.decode(data, (uint256)) : 18;
    }

    function isETH(IERC20 token) internal pure returns (bool) {
        return address(token) == address(ETH_ADDRESS);
    }
}

// File: contracts/v2/NftVault.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/utils/structs/EnumerableSet.sol";

import "./interfaces/IDepositValidator.sol";
import "./interfaces/IWithdrawValidator.sol";
import "./libs/IterableAddressMap.sol";
import "./libs/IterableLock.sol";
import "./libs/UniversalERC20.sol";
import "./BaseVault.sol";

contract NftVault is BaseVault, ReentrancyGuard {
    using EnumerableSet for EnumerableSet.AddressSet;
    using IterableLock for ItLock;
    using IterableAddressMap for ItAddressMap;
    using UniversalERC20 for IERC20;

    /// @notice Locked tokens for the nft token address + nft token id
    /// @dev key (bytes32) is the hash value of nft address and token id
    mapping(bytes32 => ItAddressMap) private _lockedTokens;

    /// @notice ERC20 token balance for the nft token
    /// @dev Key (bytes32) is the hash value of nft address, nft token id and erc20 token address
    mapping(bytes32 => ItLock) private _balances;

    /// @notice total balance for each erc20 token
    mapping(address => uint256) private _balancePerToken;

    address private _trustedForwarder;

    constructor(
        address payable treasury_,
        address depositValidator_,
        address withdrawValidator_
    ) {
        if (
            treasury_ == address(0) ||
            depositValidator_ == address(0) ||
            withdrawValidator_ == address(0)
        ) revert InvalidZeroAddress();

        _treasury = treasury_;
        _depositValidator = depositValidator_;
        _withdrawValidator = withdrawValidator_;

        _depositFee = 500; // default 5%
        _withdrawFee = 200; // default 2%
    }

    /// @notice Deposit ERC20 token with locked time for nft token + token id
    /// @param erc20Token_ ERC20 token contract address to deposit
    /// @param nftAddress_ key NFT contract address
    /// @param nftTokenId_ key NFT contract token ID
    /// @param amount_ deposit amount
    /// @param unlockAt_ the time until the deposited token is locked
    function deposit(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_,
        uint64 unlockAt_
    ) external payable nonReentrant {
        _deposit(erc20Token_, nftAddress_, nftTokenId_, amount_, unlockAt_);
        _transferToken(erc20Token_, amount_);
    }

    /// @notice Batch deposit for several nft tokens + token ids
    /// @param erc20Token_ ERC20 token contract address to deposit
    /// @param nftAddresses_ key NFT contract addresses
    /// @param nftTokenIds_ key NFT contract token IDs
    /// @param amounts_ deposit amounts
    /// @param unlockAt_ the time until the deposited tokens are locked
    function batchDeposit(
        address erc20Token_,
        address[] calldata nftAddresses_,
        uint256[] calldata nftTokenIds_,
        uint256[] calldata amounts_,
        uint64 unlockAt_
    ) external payable override nonReentrant {
        if (
            nftAddresses_.length != nftTokenIds_.length ||
            nftTokenIds_.length != amounts_.length
        ) revert InvalidArraySize();
        uint256 batchCount = nftAddresses_.length;
        uint256 batchLimit = _batchTxLimit;
        if (batchCount > batchLimit) revert Overflow256(batchLimit, batchCount);

        uint256 sumAmount;
        uint256 i;

        for (; i < batchCount; ) {
            _deposit(
                erc20Token_,
                nftAddresses_[i],
                nftTokenIds_[i],
                amounts_[i],
                unlockAt_
            );
            sumAmount += amounts_[i];

            unchecked {
                ++i;
            }
        }
        _transferToken(erc20Token_, sumAmount);
    }

    /// @notice Update balance data for the deposit operation
    function _deposit(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_,
        uint64 unlockAt_
    ) internal {
        if (unlockAt_ > block.timestamp + MAX_LOCK_TIME_BUFFER)
            revert InvalidUnlockTime();
        if (tokenInBlacklist(erc20Token_)) revert InvalidToken(erc20Token_);
        if (
            !IDepositValidator(_depositValidator).isValid(
                _msgSender(),
                nftAddress_,
                nftTokenId_
            ) ||
            (_isUseNftWhitelist && !_nftWhitelist.contains(nftAddress_)) ||
            tokenIdInBlacklist(nftAddress_, nftTokenId_)
        ) revert InvalidNft();

        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];

        // Iterate the locks and mark the expired locks as unlocked
        itLock.deposit(unlockAt_, amount_);

        ItAddressMap storage itAddressMap = _lockedTokens[
            _hash2(nftAddress_, nftTokenId_)
        ];
        itAddressMap.insert(erc20Token_);

        _balancePerToken[erc20Token_] += amount_;

        emit Deposited(
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            amount_,
            unlockAt_
        );
    }

    /// @notice Transfer deposited token to the nft vault, and the fee to the treasury account
    function _transferToken(address token_, uint256 amount_) internal {
        uint256 feeAmount = depositFee(token_, amount_);
        amount_ += feeAmount;

        // Transfer total deposit amount to the NftVault contract
        uint256 amountTransferred = IERC20(token_).universalTransferFrom(
            _msgSender(),
            address(this),
            amount_
        );
        // Fee token is not supported in the NftVault
        if (amountTransferred < amount_)
            revert InsufficientTransfer(amount_, amountTransferred);

        // Transfer fee to the treasury account
        if (feeAmount == 0) return;
        uint256 feeTransferred = IERC20(token_).universalTransfer(
            _treasury,
            feeAmount
        );

        if (feeTransferred < feeAmount)
            revert InsufficientTransfer(feeAmount, feeTransferred);
    }

    /// @notice withdraw ERC20 token from NftVault contract
    /// _msgSender() should own key NFT
    /// recipient is _msgSender()
    /// @param nftAddress_ key NFT contract address
    /// @param nftTokenId_ key NFT token ID
    /// @param erc20Token_ withdraw token contract address
    /// @param amount_ withdraw amount
    function withdraw(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_
    ) external {
        // Check the user who wants to withdraw owns this nft token id
        if (
            IERC721(nftAddress_).ownerOf(nftTokenId_) != _msgSender() ||
            !IWithdrawValidator(_withdrawValidator).isValid(
                _msgSender(),
                nftAddress_,
                nftTokenId_
            )
        ) revert Unpermitted();

        // For the withdrwal of zero amount, it is reverted
        if (amount_ == 0) revert InvalidZeroAmount();

        _withdraw(
            false,
            _msgSender(),
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            amount_
        );
    }

    /// @notice Withdraw for making payment for the Slash Protocol
    /// @param account_ Account who is going to make payment via Slash
    /// @param nftAddress_ key NFT contract address
    /// @param nftTokenId_ key NFT token ID
    /// @param erc20Token_ withdraw token contract address
    /// @param amount_ withdraw amount
    function withdrawForPayment(
        address account_,
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_
    ) external override {
        // If called via a forwarder, _msgSender() may not be a wrapper, so check with msg.sender.
        if (!_wrapperList.contains(msg.sender)) revert Unpermitted();
        bytes32 key = keccak256(abi.encodePacked(account_, msg.sender));
        if (!_withdrawForPaymentApproved[key]) revert UnapprovedWrapper();

        // Check the user who wants to make the slash payment owns this nft token id
        // We check account_, not _msgSender() because _msgSender() is Payment wrapper contract
        if (
            IERC721(nftAddress_).ownerOf(nftTokenId_) != account_ ||
            !IWithdrawValidator(_withdrawValidator).isValid(
                account_,
                nftAddress_,
                nftTokenId_
            )
        ) revert Unpermitted();

        _withdraw(
            true,
            msg.sender, // Withdraw to Payment wrapper contract
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            amount_
        );
    }

    function _withdraw(
        bool fromWrapper,
        address recipient_,
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_
    ) internal nonReentrant {
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];

        // Deposited tokens in the vault for this nft + token id + erc20 token
        uint256 tokenAmountInVault = itLock.availableAmount();

        if (amount_ > tokenAmountInVault) {
            if (fromWrapper)
                amount_ = tokenAmountInVault; // In case of the withdrawal request for the slash payment, we just withdraw available amount in the vault
            else revert TooMuchWithdrawals(tokenAmountInVault, amount_); // In case of the normal withdrawal, it is reverted
        }

        itLock.withdraw(amount_);
        _balancePerToken[erc20Token_] -= amount_;

        // If this token does not have amount left per this nft id, remove it from lockedTokens list
        if (itLock.empty())
            _lockedTokens[_hash2(nftAddress_, nftTokenId_)].remove(erc20Token_);

        uint256 feeAmount;
        if (!fromWrapper) {
            feeAmount = withdrawFee(erc20Token_, amount_);
            IERC20(erc20Token_).universalTransfer(_treasury, feeAmount);

            amount_ -= feeAmount;
        }

        IERC20(erc20Token_).universalTransfer(recipient_, amount_);

        emit Withdrawn(
            recipient_,
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            amount_,
            feeAmount
        );
    }

    /// @notice Lock the unlocked tokens directly
    /// @dev Only nft holder can relock tokens
    function lock(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 amount_,
        uint64 unlockAt_
    ) external {
        if (unlockAt_ > block.timestamp + MAX_LOCK_TIME_BUFFER)
            revert InvalidUnlockTime();
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        // Check the user is the honest holder
        if (IERC721(nftAddress_).ownerOf(nftTokenId_) != _msgSender())
            revert Unpermitted();
        itLock.lock(unlockAt_, amount_);

        emit Locked(
            _msgSender(),
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            amount_,
            unlockAt_
        );
    }

    /// @notice Relock the tokens with new lock duration
    /// @dev Only nft holder can relock tokens
    function relock(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint64 lockId_,
        uint64 unlockAt_
    ) external {
        if (unlockAt_ > block.timestamp + MAX_LOCK_TIME_BUFFER)
            revert InvalidUnlockTime();
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        // Check the user is the honest holder
        if (IERC721(nftAddress_).ownerOf(nftTokenId_) != _msgSender())
            revert Unpermitted();
        itLock.relock(lockId_, unlockAt_);

        emit Relocked(
            _msgSender(),
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            lockId_,
            unlockAt_
        );
    }

    /// @notice Unlock expired lock for the nft token & erc20 token
    /// @dev Only nft holder can unlock expired token
    function unlock(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint64 lockId_
    ) external {
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        // Check the user is the honest holder
        if (IERC721(nftAddress_).ownerOf(nftTokenId_) != _msgSender())
            revert Unpermitted();
        itLock.unlock(lockId_);

        emit Unlocked(
            _msgSender(),
            erc20Token_,
            nftAddress_,
            nftTokenId_,
            lockId_
        );
    }

    /// @notice Function for getting hash value from (address, uint256)
    function _hash2(
        address param1_,
        uint256 param2_
    ) private pure returns (bytes32) {
        return keccak256(abi.encode(param1_, param2_));
    }

    /// @notice Function for getting hash value from (address, uint256, address)
    function _hash3(
        address param1_,
        uint256 param2_,
        address param3_
    ) private pure returns (bytes32) {
        return keccak256(abi.encode(param1_, param2_, param3_));
    }

    /// @notice Return the count of tokens lokced in this nft token id
    function lockedTokenCount(
        address nftAddress_,
        uint256 nftTokenId_
    ) external view returns (uint256) {
        return _lockedTokens[_hash2(nftAddress_, nftTokenId_)].itemCount();
    }

    /// @notice Return tokens locked in this nft token id
    /// @param nftAddress_ NFT contract address to fetch
    /// @param nftTokenId_ NFT Token ID to fetch
    /// @return tokens Token list locked in this nft token id
    function lockedTokens(
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 offset_,
        uint256 count_
    ) external view returns (address[] memory tokens) {
        return
            _lockedTokens[_hash2(nftAddress_, nftTokenId_)].fetchItems(
                offset_,
                count_
            );
    }

    /// @notice View unlocked token amount for the nft token id
    function unlockedAmount(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_
    ) external view returns (uint256) {
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        return itLock.unlockedAmount;
    }

    /// @notice View data of the token locked for the nft token id
    /// @param nftAddress_ NFT contract address
    /// @param nftTokenId_ NFT token ID
    /// @param erc20Token_ Locked token
    /// @return unlockDates Array of lock end dates
    /// @return amounts Array of locked amounts
    function viewLocks(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint256 offset_,
        uint256 count_
    ) external view returns (uint64[] memory, uint256[] memory) {
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        return itLock.fetchItems(offset_, count_);
    }

    /// @notice View locked data of given lock id
    /// @param nftAddress_ NFT contract address
    /// @param nftTokenId_ NFT token ID
    /// @param erc20Token_ Locked token
    /// @param lockId_ Lock id, same as unlock date
    /// @return unlockDate Lock end date
    /// @return amount Locked amounts
    function viewLock(
        address erc20Token_,
        address nftAddress_,
        uint256 nftTokenId_,
        uint64 lockId_
    ) external view returns (uint64, uint256) {
        ItLock storage itLock = _balances[
            _hash3(nftAddress_, nftTokenId_, erc20Token_)
        ];
        return itLock.fetchItem(lockId_);
    }

    /// @notice Return total balance of the token deposited for nfts
    /// @param erc20Token_ Token address
    function balanceOf(address erc20Token_) external view returns (uint256) {
        return _balancePerToken[erc20Token_];
    }

    /// @notice Set given `forwarder_` as trustable forwarder
    function updateTrustedForwarder(address forwarder_) public onlyOwner {
        _trustedForwarder = forwarder_;
        emit UpdateTrustedForwarder(forwarder_);
    }

    function trustedForwarder() public view returns (address) {
        return _trustedForwarder;
    }

    function isTrustedForwarder(address forwarder_) public view returns (bool) {
        return forwarder_ == _trustedForwarder;
    }

    function _msgSender() internal view override returns (address sender) {
        if (isTrustedForwarder(msg.sender)) {
            // The assembly code is more direct than the Solidity version using `abi.decode`.
            /// @solidity memory-safe-assembly
            assembly {
                sender := shr(96, calldataload(sub(calldatasize(), 20)))
            }
        } else {
            return super._msgSender();
        }
    }

    function _msgData() internal view override returns (bytes calldata) {
        if (isTrustedForwarder(msg.sender)) {
            return msg.data[:msg.data.length - 20];
        } else {
            return super._msgData();
        }
    }

    /// @notice Recover tokens remained in the contract except the users deposited
    /// @dev Only owner can withdraw tokens
    /// @param token_ withdraw token contract address
    function recoverWrongToken(address token_) external onlyOwner {
        uint256 balance = IERC20(token_).universalBalanceOf(address(this));
        /// @notice It should be failed when users' deposited amount is same as total balance
        uint256 withdrawable = balance - _balancePerToken[token_];

        IERC20(token_).universalTransfer(_msgSender(), withdrawable);

        emit RecoverWrongToken(token_, withdrawable);
    }
}